


IL PRIMO SETTIMANALE DI SOFTWARE SU CARTA

PER IL TUO PERSONAL COMPUTER

Una pubblicazione della J.soft editrice


L. 1.000



PAPER soft

Anno I - N° 4

6 luglio 1984



apple



TI99



zx Spectrum



C=64

C=VIC-20

Crociera nei caraibi
Keyboard organ
Caccia all'oro
Robachase
Caverna
Clic
Spara all'alfabeto
Il topo
Spara all'alfabeto
Memory trainer

TI99VIC.it

EDITRICE  J.soft

Con la collaborazione del Gruppo Editoriale Jackson



TI-99/4A

7

Caccia all'oro

di F. Elsesser trad. e adatt. di R. Monti

TI-99/4A

15

Robachase

di G. Vaughan trad. ed adatt. di E. Re Garbagnati

isoft s.r.l.

**DIREZIONE, REDAZIONE,
AMMINISTRAZIONE**

Via Rosellini, 12
20124 MILANO
Tel. (02) 6888228

DIRETTORE RESPONSABILE:

Pietro Dell'Orco

COORDINAMENTO TECNICO:

Riccardo Paolillo

REDAZIONE

Lucio Bragagnolo
Mauro Cristub. Grizzi

GRAFICA E IMPAGINAZIONE

Luigi Chiesa
Raffaella Toffolatti

FOTOCOMPOSIZIONE:

d&b Via Vignola, 5
Tel. 02/59.85.08
20133 MILANO

CONTABILITÀ:

Giulia Pedrazzini
Flavia Bonatti

**AUTORIZZAZIONE ALLA
PUBBLICAZIONE:**

Tribunale di Milano n° 200
del 14.04.1984

STAMPA:

Elcograf
Beverate (CO)

PUBBLICITÀ

Concessionario per l'Italia e l'Estero Reina s.r.l.
Via Washington, 50
20046 MILANO

Tel. (02) 4988066 (5 linee R.A.)
Tlx. 316213 REINA I

Concessionario esclusivo per la
DIFFUSIONE in Italia e Estero:
SODIP - Via Zuretti, 25
20125 MILANO

Spedizione in abbonamento
postale Gruppo II/70
Prezzo della rivista L. 1.000

© TUTTI I DIRITTI DI
RIPRODUZIONE O TRADUZIONE
DEGLI ARTICOLI PUBBLICATI
SONO RISERVATI

HOME COMPUTER

TEXAS INSTRUMENTS

TI-99 ITALIAN USER CLUB

WWW.TI99IUC.IT

INFO@TI99IUC.IT

Thanks to :

Retroedicola.com

for this Scan.

Editing and Rework by:

TI99 Italian User Club in the year 2018

(info@ti99iuc.it)

Downloaded from www.ti99iuc.it



Caccia all'oro

Questo gioco vi introduce in un sotterraneo di un castello. Vi sono 128 camere camere su due livelli che dovrete percorrere alla ricerca di un tesoro. Potrete uscirne ricchi o non uscirne affatto. Vi ostacoleranno mostri, ladri e trabocchetti che vi faranno precipitare ad un livello inferiore o in voragini. Siccome il gioco è piuttosto lungo, è stato suddiviso in due parti. La prima parte (Programma 1) fornisce le istruzioni per il gioco e

la seconda parte (Programma 2) è il vero gioco. Si deve scrivere prima il Programma 1 e salvarlo su nastro. Senza riavvolgere il nastro, si batte il Programma 2 e, come prima, lo si registra sul medesimo nastro subito dopo il Programma 1.

Quando, dopo aver caricato il Programma 1, lo farete girare, vi verranno fornite le istruzioni per il caricamento della seconda parte.

Programma n. 1

```

90 REM TI CACCIA ALL'ORO
91 REM PROGRAMMA #1
100 CALL SCREEN(2)
110 CALL CLEAR
120 FOR C=1 TO 8
130 CALL COLOR(C,16,2)

```

```

140 NEXT C
150 PRINT "          CACCIA ALL'ORO"
:::::
160 PRINT "ISTRUZIONI....."
.....":::::
170 FOR J=1 TO 11

```

```

180 READ A,B,C
190 CALL SOUND(A,B,0,C,0)
200 NEXT J
210 DATA 125,262,131,125,294,131
,250,156,131,250,131,147,250,262
,131,100,262,131
220 DATA 100,294,123,100,311,131
,100,392,131,100,294,131,100,311
,131,200,131,262
230 CALL SOUND(200,131,0,262,3,5
23,3)
240 FOR DELAY=1 TO 500
250 NEXT DELAY
260 CALL CLEAR
270 PRINT "CACCIA ALL'ORO E'UN G
IOCO ":"DI AVVENTURA,IN CUI DEV
I ":"RISPONDERE A CERTE DOMANDE
"::
280 PRINT "MENTRE VAGHI PER LE C
AMERE E":"LE GALLERIE DI UN LAB
IRINTO":"A DUE LIVELLI"::::::
290 PRINT "PREMI QUALSIASI TASTO
"
300 CALL KEY(0,K,S)
310 IF S=0 THEN 300
320 CALL CLEAR
330 PRINT "IL PROGRAMMA"
340 PRINT "-----"
350 PRINT
360 PRINT "TI SONO DATE 1.000 MO
NETE":"D'ORO E VIENI PORTATO AL
":"LIVELLO INFERIORE DI UN "
370 PRINT "SOTTERRANEO A DUE LIV
ELLI ":"CON 128 CAMERE(64 CAMERE
":"PER LIVELLO)."::
380 PRINT "IL TUO SCOPO E' DI FU
GGIRE":"CON PIU' ORO POSSIBILE."
":"SI GUADAGNANO MONETE D'ORO":
390 PRINT "RISPONDENDO ALLA DOMA
NDA":"FATTA DAI MOSTRI CHE ABITA
NO":"IL SOTTERRANEO."::
400 PRINT "AD OGNI RISPOSTA ESAT
TA,":"TI VIENE DATO DELL'ORO IN
":"PREMIO. SE LA RISPOSTA E'":
410 PRINT "SBAGLIATA, PERDI DELL
'ORO."::
420 PRINT "PREMI QUALSIASI TASTO
"
430 CALL KEY(0,K,S)
440 IF S=0 THEN 430
450 CALL CLEAR
460 PRINT "LA DOMANDA"
470 PRINT "-----"
480 PRINT
490 PRINT "I MOSTRI TI FANNO UNA
DOMAN=":"DA PARTICOLARE MENTRE
VAGHI":"NEL SOTTERRANEO"::
500 PRINT "DOV'E' L'ORO?"::"SUG

```

```

ERIMENTO:DOVE METTERESTI"::"1.00
0 MONETE D'ORO SE TU"::"LE AVESS
I CON TE?"::
510 PRINT "PREMI QUALSIASI TASTO
PER":"CONTINUARE..."
520 CALL KEY(0,K,S)
530 IF S=0 THEN 520
540 CALL CLEAR
550 PRINT "MOSSE"
560 PRINT "-----"
570 PRINT
580 PRINT "MENTRE PERCORRI IL SO
TTERRA=":"NEO INCONTRI MOSTRI, L
ADRI,":
590 PRINT "CAMERE VUOTE, TRABOCC
HETTI,":"PORTE SEGRETE CHE PORTA
NO":"A GALLERIE NORD-SUD E EST-"
:
600 PRINT "OVEST E SUPERCHIAVI."
":"PER MUOVERTI, INSERISCI LA":
LETTERA IN PARENTESI PER LA":
610 PRINT "MOSSA O L'AZIONE DESI
DERATA"::"1- (N)ORD 2- (E)ST
":"3- (S)UD 4- (O)VEST":"5-
(F)INIRE GIOCO":
620 PRINT "6- (M)ONETE [CONTEGGI
O ORO]":"7- (G)RADINI [SALIRE SC
ALE]"::
630 PRINT "PREMI QUALSIASI TASTO
"
640 PRINT
650 CALL KEY(0,K,S)
660 IF S=0 THEN 650
670 CALL CLEAR
680 PRINT "CLASSIFICA"
690 PRINT "-----"
700 PRINT
710 PRINT "DOPO AVER COMPLETATO
IL GIO=":"CO, VIENE MOSTRATA LA
CLAS=":"SIFICA COL NUMERO DI MO
NETE":
720 PRINT "D'ORO VINTE E IL NUM
ERO DI":"TENTATIVI CHE HAI FATTO
PER":"TROVARE LA VIA D'USCITA."
:
730 PRINT "LA CLASSIFICA VARIERA
' TRA=":-500 E +1000. QUANTO PI
U'":"ALTO SARA' IL NUMERO, TANTO
":
740 PRINT "MIGLIORE SARA' LA CLA
SSIFICA.":"UN NUMERO NEGATIVO IN
DICA":"UNA CLASSIFICA SCADENTE."
:
750 PRINT "BUONA FORTUNA!!!!!!":
:"PREMI <ENTER>....."
760 CALL KEY(0,K,S)
770 IF S=0 THEN 760
780 CALL CLEAR

```

```

790 PRINT "PER GIOCARE:"::"BATTI
<NEW> E POI <OLD CS1>":::::::::::
:

```

```

800 FOR DELAY=1 TO 1500
810 NEXT DELAY

```

Programma n. 2

```

90 REM TI CACCIA ALL'ORO
91 REM PROGRAMMA #2
100 CALL CLEAR
110 FOR CC=1 TO 8
120 CALL COLOR(CC,5,11)
130 NEXT CC
140 CALL SCREEN(11)
160 FOR DDD=1 TO 500
170 NEXT DDD
180 CALL CLEAR
190 PRINT " ***CACCIA ALL'ORO!
***":::::::::::
200 PRINT "UNA FANTASTICA AVVENT
URA"::
210 FOR J=1 TO 2
220 IA=200
230 CALL SOUND(IA,131,0,262,3,52
3,3)
240 CALL SOUND(IA,147,0,294,3,58
7,3)
250 CALL SOUND(IA,131,0,311,3,62
2,3)
260 CALL SOUND(IA,156,0,311,3,62
2,3)
270 CALL SOUND(IA,131,0,294,3,58
7,3)
280 CALL SOUND(IA,156,0,294,3,58
7,3)
290 CALL SOUND(IA*2,131,0,262,3,
523,3)
300 CALL SOUND(IA,156,0,262,3)
310 CALL SOUND(IA,123,0,294,3)
320 CALL SOUND(IA,131,0,311,3)
330 CALL SOUND(IA,156,0,392,3)
340 CALL SOUND(IA,131,0,294,3)
350 CALL SOUND(IA,156,0,311,3)
360 CALL SOUND(IA*2,131,0,262,3)
370 NEXT J
380 CALL CLEAR
390 CALL SCREEN(14)
400 FOR CC=1 TO 8
410 CALL COLOR(CC,16,14)
420 NEXT CC
430 DIM P(9,9,2)
440 PRINT "VERRAI TRASPORTATO ":
:
450 PRINT "NEL SOTTERRANEO"::
460 FOR DELAY=1 TO 500
470 NEXT DELAY
480 CALL CLEAR
490 RANDOMIZE
500 MA=0

```

```

510 CA=0
520 G=1000
530 M1=1
540 K=0
550 PRINT "CHI SEI?"::
560 INPUT A$
610 CALL CLEAR
620 GOSUB 860
630 CALL SCREEN(7)
640 FOR CC=1 TO 8
650 CALL COLOR(CC,16,7)
660 NEXT CC
670 PRINT "HAI CON TE 1000 MONET
E D'ORO"::
680 GOSUB 860
690 PRINT A$;" . . . PARTI! . . .
. . ."::
700 GOSUB 860
710 GOSUB 900
720 CALL CLEAR
730 CALL SCREEN(2)
740 FOR CC=1 TO 8
750 CALL COLOR(CC,16,2)
760 NEXT CC
770 PRINT "SEI ARRIVATO NEL SOTT
ERRANEO"::
780 PRINT " LIVELLO INFERI
ORE"::
790 PRINT "INCONTRERAI MOSTRI, L
ADRI,"::
800 PRINT "E ORO..."::
810 PRINT "MA STAI ATTENTO!"::
820 PRINT "I TRABOCCHETTI POSSON
O"::"COSTARTI CARO"
830 FOR AB=1 TO 2000
840 NEXT AB
850 GOTO 1840
860 REM
870 FOR Z2=1 TO 400
880 NEXT Z2
890 RETURN
900 REM
910 FOR X=1 TO 8
920 FOR Y=1 TO 8
930 FOR Z=1 TO 2
940 P(X,Y,Z)=INT(RND*7)+1
950 NEXT Z
960 NEXT Y
970 NEXT X
980 REM
990 H=INT(RND*3)+1
1000 FOR A=1 TO 2

```

```

1010 FOR N=1 TO H
1020 X=INT(RND*8)+1
1030 Y=INT(RND*8)+1
1040 P(X,Y,A)=8
1050 NEXT N
1060 NEXT A
1070 S=INT(RND*4)+2
1080 FOR A=1 TO 2
1090 FOR N=1 TO S
1100 X=INT(RND*8)+1
1110 Y=INT(RND*8)+1
1120 P(X,Y,A)=9
1130 NEXT N
1140 NEXT A
1150 RETURN
1160 REM
1170 L1=L1-1
1180 PRINT "STAI SALENDO LA SCAL
A. . .":
1190 N=1
1200 X=INT(RND*5)
1210 IF X=2 THEN 1270
1220 CALL SOUND(5,-3,5)
1230 CALL SOUND(30,-7,20)
1240 CALL SOUND(500,-7,30)
1250 N=N+1
1260 IF N=15 THEN 1290 ELSE 1200
1270 CALL SOUND(60,-7,20)
1280 GOTO 1240
1290 REM
1300 GOSUB 860
1310 CALL SCREEN(11)
1320 FOR CC=1 TO 8
1330 CALL COLOR(CC,5,11)
1340 NEXT CC
1350 PRINT "INSERISCI LA SUPERCH
IAVE NELLA SERRATURA . . . .
. ."
1360 GOSUB 860
1370 PRINT "LA PORTA E' APERTA":
:
1380 CALL SOUND(125,131,0)
1390 CALL SOUND(125,147,0)
1400 CALL SOUND(500,165,0)
1410 CALL SOUND(250,165,0)
1420 CALL SOUND(250,165,0)
1430 CALL SOUND(250,147,0)
1440 CALL SOUND(250,165,0)
1450 CALL SOUND(750,175,0)
1460 CALL SOUND(500,165,0)
1470 FOR A=0 TO 30 STEP 5
1480 CALL SOUND(-99,698,A,1924,A
)
1490 NEXT A
1500 FOR A=0 TO 30 STEP 5
1510 CALL SOUND(-99,554,A,1527,A
)
1520 NEXT A

```

```

1530 GOSUB 860
1540 IF L1=0 THEN 1620
1550 MA=0
1560 K=0
1570 K4=INT(RND*4)+4
1580 CB=CA+K4
1590 PRINT "SEI AL LIVELLO SUPER
IORE"
1600 GOSUB 860
1610 GOTO 1900
1620 PRINT "HAI TROVATO LA STRAD
A PER ":
1630 PRINT "USCIRE DAL SOTTERRAN
EO":
1640 CALL SOUND(1000,262,2,330,2
,392,2)
1650 CALL SOUND(1000,349,2,392,2
,440,2)
1660 PRINT "HAI GUADAGNATO ";G;"
MONETE D'ORO.":
1670 FOR J=1 TO 11
1680 READ MMM,NNN,OOO
1690 CALL SOUND(MMM,NNN,0,OOO,0)
1700 NEXT J
1710 DATA 125,262,131,125,294,13
1,250,156,131,250,131,147,250,26
2,131,100,262,131
1720 DATA 100,294,123,100,311,13
1,100,392,131,100,294,131,100,31
1,131,125,262,131,200,131,262
1730 CALL SOUND(200,131,0,262,3,
523,3)
1740 GOSUB 1760
1750 GOTO 2690
1760 GG=G+100
1770 R=INT((GG*CA-7000+1)/M1)
1780 PRINT
1790 PRINT "IL TUO PUNTEGGIO E'
";R
1800 PRINT
1810 IF G<=0 THEN 5380
1820 PRINT "TI SONO OCCORSI ";M1
;"
TENTATIVI PER TROVAR
E L'USCITA":
1830 RETURN
1840 REM
1850 C=INT(RND*8)+1
1860 D=INT(RND*8)+1
1870 P(C,D,2)=1
1880 L1=2
1890 K4=INT(RND*4)+4
1900 CALL CLEAR
1910 A=P(C,D,L1)
1920 GOSUB 860
1930 ON A GOSUB 2960,3010,3060,3
060,3140,3730,3800,3860,4170
1940 PRINT
1950 IF G<=0 THEN 2570

```

```

1960 PRINT A$;" COSA DECIDI?"
1970 PRINT
1980 PRINT "(N)ORD, (S)UD, (E)ST
, (O)VEST, (G)RADINO, (M)O
NETEOPPURE (F)INE":
1990 CALL KEY(0,K,S)
2000 IF S=0 THEN 1990
2010 CALL CLEAR
2020 M1=M1+1
2030 IF M1>=140/L1 THEN 5270
2040 IF K=78 THEN 2150
2050 IF K=69 THEN 2200
2060 IF K=83 THEN 2250
2070 IF K=79 THEN 2300
2080 IF K=71 THEN 2350
2090 IF K=77 THEN 2410
2100 IF K=70 THEN 2120
2110 GOTO 1960
2120 END
2130 PRINT
2140 GOTO 1940
2150 REM
2160 IF A=7 THEN 2450
2170 IF (D-1)=0 THEN 2780
2180 D=D-1
2190 GOTO 1900
2200 REM
2210 IF A=6 THEN 2510
2220 IF (C+1)=9 THEN 2840
2230 C=C+1
2240 GOTO 1900
2250 REM
2260 IF A=7 THEN 2450
2270 IF (D+1)=9 THEN 2880
2280 D=D+1
2290 GOTO 1900
2300 REM
2310 IF A=6 THEN 2510
2320 IF (C-1)=0 THEN 2920
2330 C=C-1
2340 GOTO 1900
2350 CALL CLEAR
2360 IF A<>9 THEN 2380
2370 IF K>-1 THEN 1160
2380 PRINT "NON SEI SU UNA SCALA
":
2390 GOSUB 860
2400 REM
2410 REM
2420 CALL CLEAR
2430 PRINT "HAI ";G;" MONETE D'O
RO.":
2440 GOTO 1940
2450 REM
2460 PRINT
2470 CALL CLEAR
2480 PRINT "SEI IN UNA GALLERIA
EST-OVEST":
2490 PRINT "PUOI SOLO ANDARE A E
ST O A OVEST":
2500 GOTO 1940
2510 REM
2520 PRINT
2530 CALL CLEAR
2540 PRINT "SEI IN UNA GALLERIA
NORD-SUD":
2550 PRINT "PUOI SOLO ANDARE A N
ORD O A SUD":
2560 GOTO 2500
2570 REM
2580 GOSUB 860
2590 PRINT
2600 CALL SOUND(250,139,2)
2610 CALL SOUND(10,2000,2)
2620 CALL SOUND(250,139,2)
2630 CALL SOUND(10,2000,2)
2640 CALL SOUND(500,185,2)
2650 PRINT "HAI PERSO TUTTO L'OR
O E NON":
2660 PRINT "SEI STATO IN GRADO D
I USCIRE": "DAL SOTTERRANEO IN T
EMPO":
2670 PRINT "ANDRA'MEGLIO UN'ALTR
A VOLTA":
2680 GOSUB 1760
2690 PRINT
2700 PRINT "GIOCHI ANCORA ?":
2710 PRINT "BATTI '1'-SI '0'-NO
":
2720 INPUT AA
2730 CALL CLEAR
2740 IF AA<>1 THEN 2770
2750 CALL CLEAR
2760 GOTO 450
2770 END
2780 CALL CLEAR
2790 PRINT "SEI ALLA PARETE NORD
":
2800 CALL SOUND(-500,400,2,-4,2)
2810 PRINT "NON PUOI SUPERARLA":
:
2820 PRINT "TENTA IN UN'ALTRA DI
REZIONE.":
2830 GOTO 1940
2840 CALL CLEAR
2850 PRINT "SEI ALLA PARTE EST":
:
2860 CALL SOUND(-500,400,2,-4,2)
2870 GOTO 2810
2880 CALL CLEAR
2890 PRINT "SEI ALLA PARETE SUD"
":
2900 CALL SOUND(-500,400,2,-4,2)
2910 GOTO 2810
2920 CALL CLEAR
2930 PRINT "SEI ALLA PARETE OVES

```

```

T"::
2940 CALL SOUND(-500,400,2,-4,2)
2950 GOTO 2810
2960 REM
2970 PRINT
2980 PRINT "TI TROVI IN UNA . .
. . . ."::
2990 PRINT " . . . . STANZA V
UOTA"::
3000 RETURN
3010 REM
3020 PRINT
3030 PRINT "TI TROVI IN UNA STAN
ZA "::
3040 PRINT "VUOTA E FUMOSA "::
3050 RETURN
3060 CALL CLEAR
3070 GOSUB 5220
3080 RANDOMIZE
3090 M4=INT(RND*10)+1
3100 GOSUB 4290
3110 ON M4 GOSUB 4410,4520,4720,
4810,4410,4720,4520,4520,4410,48
10
3120 PRINT
3130 RETURN
3140 CALL CLEAR
3150 PRINT "C'E' UN LADRO IN"::"
QUESTA STANZA"::
3160 FOR I=1 TO 5
3170 CALL SOUND(500,110,2)
3180 CALL SOUND(-500,330,1)
3190 NEXT I
3200 P(C,D,L1)=2
3210 GOSUB 860
3220 G4=INT(350/L1*RND)+1
3230 RANDOMIZE
3240 Y=INT(RND*8)+1
3250 IF Y<=5 THEN 3600
3260 PRINT
3270 PRINT "TI SORPRENDE."::
3280 GOSUB 860
3290 PRINT "MENTRE TI SUPERA VEL
OCEMENTE"::
3300 PRINT "TI RUBA ";G4;" MONET
E D'ORO"::
3310 G=G-G4
3320 REM
3330 GOSUB 860
3340 PRINT "TU CERCHI LA STANZA
E TROVI"::
3350 RANDOMIZE
3360 ZZZ=INT(RND*14)+1
3370 ON ZZZ GOSUB 3390,3490,3410
,3430,3450,3470,3490,3490,3510,3
390,3530,3550,3570,3490
3380 GOTO 3590
3390 PRINT "UN MORTO"::
3400 RETURN
3410 PRINT "UNA BARA AMMUFFITA"
3420 RETURN
3430 PRINT "UNO SCARAFAGGIO"
3440 RETURN
3450 PRINT "UN MOSTRO AMMAZZATO"
3460 RETURN
3470 PRINT "UN PEZZO DI PANE"
3480 RETURN
3490 PRINT "UN FORZIERE"::
3491 GOSUB 860
3492 PRINT "TU APRI IL COPERCHIO
. . ."::
3493 GOSUB 860
3494 RANDOMIZE
3495 G54=INT(RND*2000)+1
3496 PRINT "E TROVI";G54;" MONET
E D'ORO."
3497 G=G+G54
3498 PRINT
3500 RETURN
3510 PRINT "UN TI HOME COMPUTER"
3520 RETURN
3530 PRINT "UN'ANTICA SPADA"
3540 RETURN
3550 PRINT "UNA CANDELA"
3560 RETURN
3570 PRINT "LA MASSIMA RICOMPENS
A-
1000 PEZZI D'ORO"
3580 G=G+1000
3590 RETURN
3600 PRINT "TU SORPRENDI IL LADR
O "::
3610 GOSUB 860
3620 PRINT "MENTRE SCAPPA PERDE
"::
3630 PRINT " ";G4;" MONETE D
'ORO"::
3640 PRINT "TU RACCOGLI LE MONET
E D'ORO"::
3650 G=G+G4
3660 IF M1=1 THEN 3670 ELSE 3680
3670 RETURN
3680 MA=INT(RND*4)+1
3690 IF M1<=2 THEN 3700 ELSE 371
0
3700 MA=1
3710 IF MA=1 THEN 3340
3720 RETURN
3730 CALL CLEAR
3740 PRINT
3750 GOSUB 5440
3760 REM
3770 PRINT " ENTRI IN UNA G
ALLERIA NORD-SUD"::
3780 GOSUB 5400
3790 RETURN
3800 CALL CLEAR
3810 PRINT

```



```

3820 GOSUB 5440
3830 PRINT "          ENTRI IN UNA G
ALLERIA EST-OVEST"
3840 GOSUB 5400
3850 RETURN
3860 REM
3870 CALL SOUND(250,139,2)
3880 CALL SOUND(10,2000,2)
3890 CALL SOUND(250,139,2)
3900 CALL SOUND(10,2000,2)
3910 CALL SOUND(500,185,2)
3920 PRINT "HAI MESSO IL PIEDE S
U UN  TRABOCCHETTO"::
3930 GOSUB 860
3940 TD=INT(RND*4)+1
3950 IF TD>=3 THEN 3990
3960 PRINT
3970 PRINT "MA...TI SEI SALVATO"
::
3980 RETURN
3990 IF L1=2 THEN 4100
4000 L1=L1+1
4010 K=1
4020 PRINT "SEI PRECIPITATO AL L
IVELLO INFERIORE E"::
4030 G=100
4040 GOSUB 860
4050 PRINT
4060 PRINT "HAI PERSO LA MAGGIO
PARTE  DELLE TUE MONETE D'ORO":
:
4070 PRINT "TU HAI ANCORA ";G;"
MONETE D'ORO"::
4080 PRINT "MA HAI ANCORA LA CHI
AVE"::
4090 RETURN
4100 PRINT "SEI PRECIPITATO IN U
NA    PROFONDA FOSSA"::
4110 GOSUB 860
4120 PRINT "SEI STATO FORTUNATO
!"::
4130 PRINT "NON HAI RIPORTATO DA
NNI"::
4140 GOSUB 860
4150 PRINT "MA ARRAMPICANDOTI PE
R USCIRE"::
4160 GOTO 5310
4170 PRINT "TI TROVI SU UNA SCAL
A"::
4180 PRINT "          CHE SALE"::
4190 N=1
4200 FOR F=700 TO 900 STEP 5
4210 CALL SOUND(-99,F,0)
4220 NEXT F
4230 FOR F=900 TO 700 STEP -8
4240 CALL SOUND(-99,F,0)
4250 NEXT F
4260 N=N+1
4270 IF N=2 THEN 4820 ELSE 4200
4280 RETURN
4290 CALL SOUND(250,147,0,294,0)
4300 CALL SOUND(175,147,0,294,0)
4310 CALL SOUND(75,147,0,294,0)
4320 CALL SOUND(250,147,0,294,1)
4330 CALL SOUND(175,175,0,349,0)
4340 CALL SOUND(75,165,0,330,0)
4350 CALL SOUND(175,165,0,330,0)
4360 CALL SOUND(75,147,0,294,1)
4370 CALL SOUND(175,147,0,294,0)
4380 CALL SOUND(75,131,0,162,0)
4390 CALL SOUND(500,147,0,294,0,
587,0)
4400 RETURN
4410 PRINT "FERMO! *** IO SONO D
RA"::
4420 GOSUB 4860
4430 GOSUB 4900
4440 INPUT P$
4450 IF P$="SACCO" THEN 4500
4460 IF P$="NEL SACCO" THEN 4500
4470 REM
4480 GOSUB 5080
4490 RETURN
4500 GOSUB 4930
4510 RETURN
4520 PRINT "STOP! *** IO SONO BU
S":: Downloaded from WWW.TI99iuc.it
4530 GOSUB 4860
4540 GOSUB 4900
4550 INPUT P$
4560 IF P$="NEL SACCO" THEN 4600
4570 IF P$="SACCO" THEN 4600
4580 GOSUB 5080
4590 RETURN
4600 GOSUB 4930
4610 RETURN
4620 GOSUB 900
4630 H=1
4640 O=9
4650 W=8
4660 B=0
4670 E=5
4680 R=14
4690 C=0
4700 PR=0
4710 GOTO 1840
4720 PRINT "ALT!*** IO SONO LUM!
!!"::
4730 GOSUB 4860
4740 GOSUB 4900
4750 INPUT P$
4760 IF P$="SACCO" THEN 4790
4770 GOSUB 5080
4780 RETURN
4790 GOSUB 4930
4800 RETURN

```

```

4810 PRINT "ATTENTO AI TRABOCCHETTI":
4820 CALL SOUND(500,220,0,330,0,440,0)
4830 CALL SOUND(250,330,0,440,0,550,0)
4840 REM
4850 RETURN
4860 PRINT
4870 PRINT "NON PUOI PASSARE FINCHE'":
4880 PRINT "NON MI DICI DOV'E' L'ORO":
4890 RETURN
4900 REM
4910 RETURN
4920 CALL CLEAR
4930 PRINT "BENE!DEVI VENIR RICOMPENSATO"
4940 PRINT
4950 G4=INT(400/L1*RND)+25
4960 K=1
4970 G=G+G4
4980 GOSUB 860
4990 PRINT "VINCI ";G4;" MONETE D'ORO":
5000 GOSUB 5180
5010 P(C,D,L1)=1
5020 CA=CA+1
5030 IF K=1 THEN 5040 ELSE 5050
5040 RETURN
5050 IF CA=K4 THEN 5170
5060 IF L1=1 THEN 5290
5070 RETURN
5080 PRINT
5090 PRINT "TU MENTI!":
5100 PRINT "NESSUNA RICOMPENSA PER TE":
5110 PRINT
5120 G4=INT(350/L1*RND)+1
5130 G=G-G4
5140 GOSUB 860
5150 PRINT "IO VINCO ";G4;" MONETE D'ORO":
5160 RETURN
5170 GOSUB 860
5180 K=1
5190 PRINT "HAI TROVATO LA SUPERCHIAVE":


```

```

5200 GOSUB 860
5210 RETURN
5220 PRINT "HAI DISTURBATO UN MOMENTO NELLA SUA CAMERA":
5230 GOSUB 860
5240 PRINT "ED EGLI DICE: . . .
. . . ":
5250 GOSUB 860
5260 RETURN
5270 GOSUB 5180
5280 GOTO 2040
5290 IF CA=CB THEN 5170
5300 RETURN
5310 G=100
5320 GOSUB 860
5330 PRINT
5340 PRINT "                HAI PER SO":
5350 PRINT "LA MAGGIOR PARTE DELLE TUE MONETE D'ORO":
5360 PRINT "HAI ANCORA ";G;" MONETE D'ORO":
5370 RETURN
5380 REM
5390 RETURN
5400 PRINT "LA PORTA SI CHIUDE E SI BLOCCA DIETRO DI TE":
5410 CALL SOUND(1000,8207,30,880,3,30,500,30,-8,5)
5420 GOSUB 860
5430 RETURN
5440 PRINT "APRI UNA PORTA SEGRETA E...":
5450 CALL SOUND(1000,8207,30,880,3,30,500,30,-8,5)
5460 GOSUB 860
5470 RETURN

```

Visita il Sito
www.ti99iuc.it!



PAPER
4 soft

ritaglia
e
conserva
ti
attendono
grosse
sorprese!

Robochase

Mentre stai effettuando un'operazione di ricognizione nella galassia per controllare la veridicità delle voci riguardanti strani avvenimenti nel settore 47, improvvisamente la tua astronave viene fatta prigioniera da strani androidi.

È così che vieni condotto sull'asteroide 245/7b, dove incontri Von Schatzer, lo scienziato pazzo, imbestialito perché un intruso ha tentato di scoprire i suoi piani di conquista dell'universo.

Dopo un sommario processo, viene emessa la sentenza di morte nei tuoi confronti; ma il metodo di esecuzione è, pazzo come colui che lo ha ideato: vieni gettato, disarmato, in una stanza piena di barriere elettrificate e di robot programmati per gettarsi su di te disintegrandoti. È la fine? Ben presto ti accorgi che l'unica via di salvezza è quella di attirare verso di te i robot, stando in prossimità di una barriera e scansandoti all'ultimo momento, in modo da eliminare i robot facendoli sbattere contro di essa. Inoltre ti accorgi-

che Von Schatzer ti ha lasciato la tua cintura teletrasportatrice, con la quale puoi materializzarti da un luogo all'altro della stanza; l'unico guaio è che le sue batterie hanno una scarsa autonomia; ma è inconveniente che gli elimini ben presto grazie alla scoperta che gli accumulatori per le barriere elettrificate (a forma di diamante) possono essere utilizzati per un'ulteriore ricarica del tuo teletrasporto. Quando riesci a distruggere tutti i robot blu, ti trovi catapultato in un'altra stanza dove incontri maggiori difficoltà contro i robot-killers. Finché non arriva Spunky il marziano, inviato da Von Schatzer, per distruggere tutti i ricaricatori e le barriere, e qualche robot, tanto per rendere il gioco interessante... Eliminandolo avrai una ricarica in più per la cintura. Riuscirai a liberarti dello scienziato pazzo?

(Ogni ulteriore istruzione per questo avvincente gioco in Basic per il tuo TI-99 è inclusa nel listato... Buona fortuna!)

```

100 REM *****
110 REM * ROBOCHASE *
120 REM *****
130 REM
140 REM TI99/4A BASIC
150 REM CREAZIONE TITOLI
    DI TESTA
160 RANDOMIZE
170 CALL SCREEN(16)
180 CALL CLEAR
190 DIM Z$(4)
200 DATA FFFFFFFFFFFFFFFF,010307
    0F1F3F7FFF,FF7F3F1F0F070301,80C0
    E0F0F8FCFEFF,FFFEFCF8F0E0C080
210 DATA FCFCFCFCFCFCF,3F3F3F
    3F3F3F3F3F
220 DATA E0F0F8F0E0F0F8,F8F8D8D8
    D8F8F8,F0D8D8F0D8D8F0,78F0E0E0E0
    F078,8888D8F8D88888
230 DATA 2070D8D8F8D888,70D8C070
    18D870,F8F0E0F8E0F0F8

```

```

240 DATA FCFCCCCCCCCFCFC,383818
    1818181818,FCFC0CFCC0FCFC,FCFC
    0CFCF0CFCF,CCCCCFCF0C0C0C
250 DATA FCFCC0FCFC0CFCF,FCFCC0
    FCFCCFCFC,FCFC0C0C0C0C0C0C,FCFC
    CCFCFCCCFCF,FCFCCFCFC0C0C0C
260 DATA 8040,2010,0804,0201
270 FOR X=33 TO 39
280 READ A$
290 CALL CHAR(X,A$)
300 NEXT X
310 FOR X=104 TO 111
320 READ A$
330 CALL CHAR(X,A$)
340 NEXT X
350 FOR X=48 TO 57
360 READ A$
370 CALL CHAR(X,A$)
380 NEXT X
390 READ A$,B$,C$,D$
400 Z$(1)=A$&B$&C$&D$

```

```

410 Z$(2)=D$&A$&B$&C$
420 Z$(3)=C$&D$&A$&B$
430 Z$(4)=B$&C$&D$&A$
440 CALL CHAR(97,Z$(1))
450 CALL CHAR(98,Z$(4))
460 CALL COLOR(9,9,16)
470 PRINT "aaaaaaaaaaaaaaaaaaaaaa
aaaaaaaaaaaaaaaaaaaaaaaaaaaaaa
aaa"
480 A$=CHR$(34)
490 PRINT "a!$ !! !$ !! !% &' ";
A$;"$ ";A$;"$ !%b"
500 PRINT "a!% &' !% &' ! !! &'
!$ !$b"
510 PRINT "a!$ &' !$ &' ! !! !!
#! !%b"
520 PRINT "a&' !! !% !! !$ &' %#
#% !$b"
530 PRINT "bbbbbbbbbbbbbbbbbbbbbb
bbbbbbbbbbbbbbbbbbbbbbbbbbbbbb
bbb"
540 PRINT :::" <PREMI UN T
ASTO> ":" <O -I- PER LE IST
RUZIONI> ":::::
550 X=3
560 Y=1
570 CALL COLOR(1,X,1)
580 CALL CHAR(97,Z$(Y))
590 CALL CHAR(98,Z$(5-Y))
600 CALL KEY(3,K,S)
610 IF S<>0 THEN 700
620 X=X+1
630 Y=Y+1
640 IF Y<5 THEN 660
650 Y=1
660 IF X<13 THEN 570
670 X=3
680 GOTO 570
690 REM INSERIMENTO
VARIABILI
700 R=10
710 D=1
720 S=0
730 LE=1
740 BA=24
750 DI=6
760 CALL CLEAR
770 CALL CHAR(126,"000008181C3C3
E7E")
780 CALL CHAR(119,"81423C247E243
CFF")
790 CALL CHAR(128,"00107C3838282
800")
800 CALL CHAR(136,"38107C7C7C6C6
C6C")
810 CALL CHAR(144,"007E7E7E7E7E7
E00")
820 CALL CHAR(152,"10387CFE7C381

```

```

")
830 CALL SCREEN(15)
840 CALL COLOR(1,2,1)
850 CALL COLOR(13,3,16)
860 CALL COLOR(14,5,15)
870 CALL COLOR(15,7,16)
880 CALL COLOR(16,16,15)
890 CALL COLOR(11,3,1)
900 DIM A(10)
910 DIM B(10)
920 IF K=73 THEN 3350
930 PRINT " hijiklmno"
940 PRINT " !!!!!!!!!!!!!!!!!!!!!
!!!"
950 FOR X=1 TO 20
960 PRINT " !
!"
970 NEXT X
980 PRINT " !!!!!!!!!!!!!!!!!!!!!
!!!"
990 FOR X=1 TO BA
1000 CALL HCHAR(INT(RND*20+3),IN
T(RND*20+6),144)
1010 NEXT X
1020 FOR X=1 TO DI
1030 CALL HCHAR(INT(RND*20+3),IN
T(RND*20+6),152)
1040 NEXT X
1050 FOR X=1 TO 10
1060 A(X)=INT(RND*20+6)
1070 B(X)=INT(RND*20+3)
1080 CALL GCHAR(B(X),A(X),CH)
1090 IF CH<>32 THEN 1060
1100 CALL HCHAR(B(X),A(X),136)
1110 NEXT X
1120 BU=0
1130 IF LE<4 THEN 1220
1140 BU=1
1150 NX=1
1160 NY=1
1170 BX=INT(RND*19+6)
1180 BY=INT(RND*19+3)
1190 CALL GCHAR(BY,BX,CH)
1200 IF CH<>32 THEN 1170
1210 CALL HCHAR(BY,BX,119)
1220 Q=INT(RND*20+6)
1230 W=INT(RND*20+3)
1240 CALL GCHAR(W,Q,CH)
1250 IF CH<>32 THEN 1220
1260 RT=0
1270 CALL HCHAR(1,28,48+D)
1280 GOSUB 2530
1290 CALL HCHAR(W,Q,32)
1300 REM MOVIMENTO OMINO
1310 CALL KEY(3,I,T)
1320 CALL KEY(1,X,Y)
1330 IF (X<>18)+(I=81) THEN 1360
1340 I=32

```

```

1350 GOTO 1420
1360 CALL JOYST(1,X,Y)
1370 IF (X=0)*(Y=0) THEN 1420
1380 IF X=4 THEN 1440
1390 IF X=-4 THEN 1470
1400 IF Y=4 THEN 1500
1410 IF Y=-4 THEN 1530
1420 IF (I=32)*(D>0) THEN 2370
1430 IF I<>47 THEN 1460
1440 Q=Q+1
1450 GOTO 1540
1460 IF I<>80 THEN 1490
1470 Q=Q-1
1480 GOTO 1540
1490 IF I<>81 THEN 1520
1500 W=W-1
1510 GOTO 1540
1520 IF I<>65 THEN 1540
1530 W=W+1
1540 CALL GCHAR(W,Q,CH)
1550 IF CH=136 THEN 2010
1560 IF CH=144 THEN 2050
1570 IF CH=152 THEN 2480
1580 IF CH=33 THEN 2050
1590 IF CH=126 THEN 1970
1600 IF CH=119 THEN 3180
1610 CALL HCHAR(W,Q,128)
1620 CALL SOUND(50,300,0,600,0,1
200,0)
1630 REM MOVIMENTO ROBOTS
1640 IF BU=1 THEN 3040
1650 FOR X=1 TO 10
1660 IF A(X)=0 THEN 1940
1670 CALL HCHAR(B(X),A(X),32)
1680 CH=.24+(LE*.03)
1690 IF CH<=.43 THEN 1710
1700 CH=.43
1710 IF RND>.5+(CH/2) THEN 1780
1720 IF A(X)=Q THEN 1780
1730 IF A(X)<Q THEN 1760
1740 A(X)=A(X)-1
1750 GOTO 1770
1760 A(X)=A(X)+1
1770 IF (RND>CH/(.5+(CH/2)))+(B(
X)=W) THEN 1830
1780 IF B(X)=W THEN 1730
1790 IF B(X)<W THEN 1820
1800 B(X)=B(X)-1
1810 GOTO 1830
1820 B(X)=B(X)+1
1830 CALL GCHAR(B(X),A(X),CH)
1840 IF CH=128 THEN 2010
1850 IF (CH=136)*(LE<5) THEN 2920
1860 IF CH=32 THEN 1930
1870 A(X)=0
1880 S=S+25
1890 CALL SOUND(50,-5,0)
1900 R=R-1
1910 IF R=0 THEN 2100
1920 GOTO 1940
1930 CALL HCHAR(B(X),A(X),136)
1940 NEXT X
1950 GOTO 1290
1960 REM MESSAGGI DI MORTE
1970 GOSUB 2780
1980 CALL CLEAR
1990 PRINT "TI SEI SCHIANTATO CO
NTRO UN":"MUCCHIO DI ROTTAMI":"N
EL LIVELLO";LE::
2000 GOTO 2660
2010 GOSUB 2780
2020 CALL CLEAR
2030 PRINT "SEI STATO CATTURATO
DA UN":"ROBOT. NEL LIVELLO";LE:
:
2040 GOTO 2660
2050 GOSUB 2780
2060 CALL CLEAR
2070 PRINT "SEI RIMASTO ELETTRIF
ICATO ":"NEL LIVELLO";LE::
2080 GOTO 2660
2090 REM PASSAGGIO DI LIVELLO

2100 CALL COLOR(15,7,15)
2110 S=S+125
2120 FOR X=1 TO 5
2130 CALL SCREEN(7)
2140 CALL SOUND(-1000,110,0,220,
0,400,0,-7,0)
2150 CALL SCREEN(15)
2160 CALL SOUND(-1000,300,0,600,
0,800,0,-7,0)
2170 NEXT X
2180 CALL SOUND(-1,40000,30)
2190 R=10
2200 LE=LE+1
2210 IF BA<=12 THEN 2230
2220 BA=BA-4
2230 IF DI<=3 THEN 2250
2240 DI=DI-1
2250 D=INT(D/2)
2260 CALL HCHAR(1,28,48+D)
2270 GOSUB 2530
2280 CALL CLEAR
2290 CALL COLOR(15,7,16)
2300 PRINT "PASSAGGIO AL LIVELLO
";LE::
2310 PRINT "PUNTEGGIO ATTUALE:
";S:::
2320 FOR JJ=1 TO 1000
2330 NEXT JJ
2340 CALL CLEAR
2350 GOTO 930
2360 REM TELETRASPORTO
2370 GOSUB 2530
2380 D=D-1

```

```

2390 S=S-50
2400 CALL HCHAR(1,29,48+D)
2410 CALL HCHAR(W,Q,32)
2420 Q=INT(RND*20+6)
2430 W=INT(RND*20+3)
2440 CALL GCHAR(W,Q,CH)
2450 IF CH<>32 THEN 2420
2460 GOSUB 2530
2470 GOTO 1290
2480 D=D+1
2490 CALL HCHAR(1,29,48+D)
2500 S=S+75
2510 CALL SOUND(50,-3,0,700,0)
2520 GOTO 1610
2530 CALL HCHAR(W,Q,128)
2540 FOR X=1 TO 5
2550 FOR Y=1 TO INT(RND*20+2)
2560 NEXT Y
2570 CALL COLOR(13,16,16)
2580 CALL SOUND(10,500,0)
2590 FOR Y=1 TO INT(RND*20+2)
2600 NEXT Y
2610 CALL COLOR(13,6,16)
2620 CALL SOUND(10,200,0)
2630 NEXT X
2640 RETURN
2650 REM FINE GIOCO
2660 PRINT "HAI TOTALIZZATO ";S;
"P.TI":
2670 PRINT "ANCORA UNA SFIDA.(S/N
)"?"
2680 T=0
2690 CALL KEY(3,K,S)
2700 IF K=83 THEN 700
2710 IF K=78 THEN 2770
2720 T=T+1
2730 IF T<250 THEN 2690
2740 CALL CLEAR
2750 CALL SCREEN(16)
2760 GOTO 440
2770 END
2780 CALL HCHAR(W,Q,128)
2790 FOR X=1 TO 5
2800 FOR Y=1 TO 10
2810 NEXT Y
2820 CALL COLOR(13,16,7)
2830 CALL SOUND(-1000,110,0,120,
0,130,0,-6,0)
2840 FOR Y=1 TO 20
2850 NEXT Y
2860 CALL COLOR(13,6,16)
2870 CALL SOUND(-1000,220,0,240,
0,260,0,-7,0)
2880 NEXT X
2890 CALL SOUND(-8,40000,0)
2900 RETURN
2910 REM ROUTINES PER
I ROBOTS

```

```

2920 FOR Y=1 TO 10
2930 IF (A(X)=A(Y))*(B(X)=B(Y))*
(X<>Y) THEN 2950
2940 NEXT Y
2950 S=S+50
2960 R=R-2
2970 A(Y)=0
2980 CALL HCHAR(B(X),A(X),32)
2990 IF LE>2 THEN 3010
3000 CALL HCHAR(B(X),A(X),126)
3010 A(X)=0
3020 CALL SOUND(300,300,0,-3,0)
3030 GOTO 1910
3040 CALL HCHAR(BY,BX,32)
3050 BX=BX+NX
3060 IF (BX<>25)*(BX<>6) THEN 308
0
3070 NX=-NX
3080 BY=BY+NY
3090 IF (BY<>22)*(BY<>3) THEN 311
0
3100 NY=-NY
3110 CALL GCHAR(BY,BX,CH)
3120 IF CH=32 THEN 3160
3130 IF CH=136 THEN 3270
3140 IF CH=128 THEN 3180
3150 CALL SOUND(200,440,0,880,0,
523,0)
3160 CALL HCHAR(BY,BX,119)
3170 GOTO 1650
3180 S=S+100
3190 D=D+1
3200 CALL HCHAR(1,29,48+D)
3210 CALL HCHAR(W,Q,128)
3220 CALL SOUND(80,1397,0)
3230 CALL SOUND(80,1319,0)
3240 CALL SOUND(80,1397,0)
3250 BU=0
3260 GOTO 1650
3270 FOR X=1 TO 10
3280 IF (A(X)=BX)*(B(X)=BY) THEN
3300
3290 NEXT X
3300 A(X)=0
3310 R=R-1
3320 IF R=0 THEN 2100
3330 GOTO 3150
3340 REM ISTRUZIONI
3350 CALL CLEAR
3360 CALL SOUND(200,440,0,880,0,
523,0)
3370 PRINT " hijiklmno"
3380 PRINT :::
3390 PRINT " SEI STATO IMPRIGION
ATO IN ":"UNA STANZA DA UNO SCIE
NZIATO ":"PAZZO.":" PER ELIMINART
I, TI HA":
3400 PRINT "SGUINZAGLIATO CONTRO

```

```

I SUOI": "ROBOT-KILLERS."::"PER
SALVARTI DEVI CONDURLI A": "SCHI
ANTARSI CONTRO LE ":
3410 PRINT "BARRIERE ELETRIFICA
TE.":::
3420 GOSUB 3670
3430 PRINT "      hijiklmno"::
:
3440 PRINT "      ";CHR$(128);"--
TU "::::"      ";CHR$(136);"-- UN RO
BOT"::::"      ";CHR$(144);"-- UNA B
ARRIERA":::
3450 PRINT "      ";CHR$(152);"--
UN RICARICATORE PER": "      TE
LETRASPORTO":::"      w-- SPUNKY IL
MARZIANO":::
3460 GOSUB 3670
3470 PRINT "      hijiklmno"::
:
3480 PRINT "      PUOI EFFETTUARE I
TUOI": "MOVIMENTI USANDO LA TASTI
ERA": "OPPURE I JOYSTICK.":::
3490 PRINT "      Q - SU      P - S
INISTRA": "      A - GIU      / - DES
TRA":::
3500 PRINT "SONO I TASTI DIREZIO
NALI.":::::
3510 GOSUB 3670
3520 PRINT "      hijiklmno"::
:
3530 PRINT "      PUOI TELETRASPORTAR
TI IN UN": "ALTRO PUNTO DELLA STA
NZA": "PREMENDO LA BARRA O IL ":
3540 PRINT "TASTO ROSSO DEL JOYS
TICK.":::"      COMUNQUE PUOI EFFETTU
ARE": "SOLTANTO UN NUMERO LIMITAT
O":
3550 PRINT "DI TELETRASPORTI"::"
L INDICATORE DEI TELE-": "TRASPOR
TI POSSIBILI E POSTO": "NELL ANGO
LO SUPERIORE DESTRO": "DELLO SCHE
RMO.":::::
3560 GOSUB 3670
3570 PRINT "      hijiklmno"::
:
3580 PRINT "      PUOI ASSICURARTI DE
I": "TELETRASPORTI SUPPLEMENTARI"
: "CORRENDO CONTRO I RICARICA-": "
TORI A FORMA DI DIAMANTE.":::
3590 PRINT "      ELIMINANDO SPUNKY D
ISPORRAI": "DI UN ALTRA CARICA PE
R": "TELETRASPORTO.":::::
3600 GOSUB 3670
3610 PRINT "      hijiklmno"::
:
3620 PRINT "      QUANDO AVRAI ELIMIN
ATO I":::"      DIECI ROBOT DI UN LIVEL
LO ": "PASSERAI AL SUCCESSIVO."::
:::
3630 PRINT "      SPUNKY NON APPARIRA
FINO AL":::"      QUARTO LIVELLO."
:::"      IN BOCCA AL LUPO-----"
:::
3640 GOSUB 3670
3650 CALL CLEAR
3660 GOTO 930
3670 PRINT ::" <PREMI UN TASTO>"
3680 CALL KEY(3,K,ST)
3690 IF ST=0 THEN 3680
3700 CALL CLEAR
3710 CALL SOUND(200,440,0,880,0,
523,0)
3720 RETURN

```

SONO IN EDICOLA



Possiedi un Sinclair Spectrum o uno ZX81 e vuoi saperne di più? Vuoi disporre di nuovi programmi, giochi e idee per utilizzare al meglio il tuo piccolo gioiello? Corri in edicola e acquista SUPERSINC per dare una nuova carica al tuo home computer. SUPERSINC, in edicola tutti i mesi, è la rivista nuova e indispensabile per chi possiede un computer Sinclair.

SUPERVIC è il "carburante" per il tuo VIC 20 e il tuo C 64. SUPERVIC è la nuova rivista pensata e realizzata per utilizzare ancora meglio il tuo personal computer. SUPERVIC ti propone, ogni mese giochi, utility, software tools, notizie, prezzi. E tante idee nuove! Il pieno di software si fa in edicola con SUPERVIC.

 **J.soft**

Un marchio di **TechnoClub** s.r.l.

Via Rosellini 12, 20124 Milano tel. 6880951 TLX 333436

...Idee nuove per il tuo personal.